

Universidad de Buenos Aires  
FACULTAD DE INGENIERIA

*Teoria de Lenguaje*

HASKELL

*1º cuatrimestre 2015*

Sebastian Gavrilov, 96252  
Pablo Hazan, 96522  
Esteban Bellegarde,

13 de julio de 2015

# Índice

<b>1. Inferencia de tipos</b>	<b>1</b>
<b>2. Recursión</b>	<b>3</b>
<b>3. Pereza</b>	<b>4</b>
<b>4. Garbage Collection</b>	<b>4</b>
<b>5. Funciones de orden superior</b>	<b>5</b>
5.1. Funciones como parametro . . . . .	5
5.2. Funciones como resultado . . . . .	6
5.3. Plieges (folds) . . . . .	6
5.4. Funciones Anónimas (lambdas) . . . . .	6
<b>6. Functores</b>	<b>7</b>
6.1. Leyes de los Functores . . . . .	7
6.2. Functores Aplicativos . . . . .	8
<b>7. Mónadas</b>	<b>8</b>
7.1. Las Leyes de las Mónadas . . . . .	9
7.2. Mónadas comunes . . . . .	10
7.3. Mónada Maybe . . . . .	10
7.3.1. Either . . . . .	10
7.4. Mónada Error . . . . .	10
7.5. Mónada List . . . . .	10
7.6. Mónada IO . . . . .	11
7.7. Mónada State . . . . .	11
<b>8. Zippers</b>	<b>12</b>
<b>9. Concurrencia</b>	<b>13</b>
9.1. Threads . . . . .	13
9.2. Comunicación entre hilos . . . . .	13
9.3. Comunicación por canales . . . . .	14
9.4. Aclaraciones y problemas . . . . .	14
<b>10.Paralelismo</b>	<b>14</b>
<b>11.Software Transactional Memory</b>	<b>15</b>

## 1. Inferencia de tipos

Haskell permite definir funciones sin decir de que tipos son estas, esto, muy lejos de querer decir que Haskell no usa un tipado fuerte, es debido a que realiza inferencia de tipos. Un puede escribir una función que sume dos números como:

```

1 sumar :: (Num a) => a -> a -> a
2 sumar x y = x + y

```

Donde la primer linea indica que las variables a son de la clase de tipos Num. Si en GHCI usamos el comando `:t` para pedir el tipo de la funcion sumar tendremos:

```
1  :t sumar
2  sumar :: (Num a) => a -> a -> a
```

Este resultado era evidente, ya que en la propia definición lo habíamos indicado. Pero Haskell permite definir la función sumar de la siguiente manera:

```
1  sumar x y = x + y
```

Al utilizar el comando `:t` en GHCI obtendremos la misma respuesta que antes, aunque esta vez no le hayamos indicado al lenguaje los tipos en la definición. Este resultado proviene de que Haskell realiza inferencia de tipos basandose en el algoritmo de Hindley Milner o Damas Milner. Este algoritmo sigue seis reglas lógicas:

$$\begin{array}{c}
 \frac{x : \sigma \in \Gamma}{\Gamma \vdash x : \sigma} \quad [\text{Var}] \\
 \\
 \frac{\Gamma \vdash e_0 : \tau \rightarrow \tau' \quad \Gamma \vdash e_1 : \tau}{\Gamma \vdash e_0 e_1 : \tau'} \quad [\text{App}] \\
 \\
 \frac{\Gamma, x : \tau \vdash e : \tau'}{\Gamma \vdash \lambda x . e : \tau \rightarrow \tau'} \quad [\text{Abs}] \\
 \\
 \frac{\Gamma \vdash e_0 : \sigma \quad \Gamma, x : \sigma \vdash e_1 : \tau}{\Gamma \vdash \text{let } x = e_0 \text{ in } e_1 : \tau} \quad [\text{Let}] \\
 \\
 \frac{\Gamma \vdash e : \sigma' \quad \sigma' \sqsubseteq \sigma}{\Gamma \vdash e : \sigma} \quad [\text{Inst}] \\
 \\
 \frac{\Gamma \vdash e : \sigma \quad \alpha \notin \text{free}(\Gamma)}{\Gamma \vdash e : \forall \alpha . \sigma} \quad [\text{Gen}]
 \end{array}$$

Para explicar que quieren decir estas seis reglas primer es necesario entender la sintaxis y el significado de alguno de los símbolos.

Lo primero es la barra horizontal, esta viene a cumplir el rol de una implicación lógica, o sea, si el antecedente, lo de arriba, es verdadero entonces, necesariamente, el consecuente, lo de abajo, también lo es. Cuando en la parte superior hay N hipótesis, para asegurar el consecuente es necesario que se cumplan las N hipótesis, es como si las separaciones por espacio estuviesen indicando una operación and lógica. Lo siguiente son los dos puntos `;;`; estos indican de que tipo es una expresión, por ejemplo `x : Int` indica que `x` es del tipo `int`. El contexto se representa con  $\gamma$ . Para indicar que algo se encuentra en un determinado contexto usamos  $\in$ .  $\vdash$  es básicamente que se puede demostrar algo, lo precede el contexto necesario para demostrar lo que quiere. La coma `;` es utilizada para ampliar el contexto, una unión, si entendemos al contexto como el conjunto de referencias a variables de determinados tipos, donde se pisa el elemento si este ya se encontraba en el contexto original. Por ultimo, el símbolo  $\sqsubseteq$  indica una especie de inclusión, en realidad es mas bien una herencia, donde el tipo de la izquierda es un subtipo del de la derecha. Con todos la notación clara es sencillo entender las seis reglas. Por Ejemplo, la segunda indica que si en nuestro contexto tenemos una expresión `e0` que toma un elemento del tipo  $\tau$  y devuelve otro que es del tipo  $\tau'$ , y en ese mismo entorno tenemos una segunda expresión que es del tipo  $\tau$ , entonces

podemos decir que en ese mismo entorno que la expresión  $e_0 \ e_1$  es del tipo  $\tau'$ . La quinta nos dice que si en un entorno la expresión  $e$  es del tipo  $\sigma'$  y que  $\sigma'$  es un subtipo de  $\sigma$  entonces la expresión  $e$  es del tipo  $\sigma$ .

El algoritmo de inferencia de tipos usa estas reglas de forma recursiva. Realiza un ajuste de patrones, donde mira la expresión que tiene, y en cual de los consecuentes encaja mejor, una vez encontrado va a los antecedentes y mira que necesita para que estos se cumplan, si con la información actual es suficiente entonces ha terminado de inducir los tipos, en caso contrario toma estos antecedentes que aun no tiene determinados y vuelve a realizar el mismo ajuste de patrón, y así va infiriendo en cada sub-expresión los tipos necesarios. Cuando encuentra que la información alcanzada es suficiente para cumplir TODAS las hipótesis la inferencia esta completa.

Si hacemos el seguimiento del como funciona este algoritmo en la función sumar que tenemos más arriba, y utilizando la curificación de Haskell tenemos que la función es de la forma

```
1 sumar x y = x + y
```

que es lo mismo que

```
1 sumar x y = (+) x y
```

para ajustar el patrón tomamos

```
1 e0 = (+) x
```

```
2 e1 = y
```

esto encaja en el segundo patrón, donde se realiza  $e_0 \ e_1$  que es del tipo  $\tau'$ , para esto se induce que  $e_0$  es del tipo  $\tau \rightarrow \tau'$  y que  $e_1$  es tipo  $\tau'$ . Hasta ahora solo sabemos que  $y$  es del tipo  $\tau$ . Luego hay que ajustar el patrón  $e_0$ . Donde volvemos a caer en el segundo caso. Entonces tenemos que :

```
1 e'0 = (+)
```

```
2 e'1 = x
```

Con  $e'0 \ e'1$  es del tipo  $\tau \rightarrow \tau'$  Concluimos que, dado que  $e'0$  es del tipo  $\tau_1 \rightarrow \tau_1'$  y que  $e'1$  es del tipo  $\tau_1$ , como  $e'0 \ e'1 = e_0$  entonces  $e_0 : \tau_1' = \tau \rightarrow \tau'$  Hasta aquí tenemos que

```
1 y : τ
```

```
2 x : τ1
```

```
3 (+) : τ1 → τ → τ'
```

Además, como  $(+) : (\text{Num } a) \Rightarrow a \rightarrow a \rightarrow a$  Por la primera regla tenemos que  $\tau_1 \rightarrow \tau \rightarrow \tau' = a \rightarrow a \rightarrow a$  O sea

```
1 τ1 = a
```

```
2 τ = a
```

```
3 τ' = a
```

Y, finalmente, usando la quinta regla tenemos que

```
1 τ1 ⊆ a
```

```
2 τ ⊆ a
```

```
3 τ' ⊆ a
```

Finalmente solo resta recomponer todo, usando que  $\text{sumar} = e'0 \ e'1 \ e_1$  El tipo de sumar es  $(\text{Num } a) \Rightarrow a \rightarrow a \rightarrow a$ .

## 2. Recursión

La recursión o recursividad es la forma de especificar una función basandose en su propia definición. Es una parte muy importante de Haskell. Una función es recursiva cuando una parte de su definición incluye a la propia función. Necesita por lo menos un *caso base* que no hace llamado recursivo para que exista una condición límite.

Este ejemplo muestra una función de factorial recursiva, separando claramente el caso base.

```

1 factorial 0 = 1
2 factorial n = n * factorial (n - 1)

```

Muchas de las funciones comunes en Haskell se pueden definir de forma recursiva, por ejemplo el *length*, la función que devuelve el número de elementos de una lista:

```

1 length [] = 0
2 length (x:xs) = 1 + length xs

```

Recursión es usada para definir casi todas las funciones de manejo de números y listas, sin embargo en la práctica no es de usarse tan seguido: la recursividad está abstraída por las funciones de las librerías de Haskell, permitiendo al programador razonar sus problemas en más alto nivel. Por ejemplo, la función de factorial de ejemplo escrita anteriormente se puede definir de la siguiente manera:

```

1 factorial n = product [1..n]

```

### 3. Pereza

Una característica destacable de Haskell es que es *perezoso* (o *no estricto*). Esto significa que nada se va a evaluar hasta que sea directamente necesario - la evaluación queda diferida hasta que el resultado es requerido por otra computación.

El siguiente es un ejemplo de la pereza de Haskell:

```

1 let (a, b) = (length [1..5], reverse "hola mundo") in 1 + 1

```

Como la expresión después del *in* no utiliza los valores de *a* y *b*, estos quedan sin evaluar, ya que no son necesarios. También pueden no ser necesarios por completo, o ser *parcialmente* requeridos:

```

1 let (a, b) = (length [1..5], reverse "hola mundo")
2 'o':ss = b

```

Como solo es necesaria la primera letra para concluir el matcheo de patrones, la evaluación es parcial.

Las funciones en Haskell pueden ser perezosas o estrictas. Podemos analizar si una función es perezosa o estricta pasándole **undefined** y viendo si su ejecución falla. Esto se debe a que en Haskell la evaluación forzada del **undefined** siempre termina en un error.

```

1 let (x, y) = undefined in x -- Error!
2 length [undefined, undefined, undefined]
3 -- No hay error, length es perezoso

```

La evaluación perezosa tiene muchas ventajas, como la reutilización de código, posibilidad de generar estructuras de datos infinitas y definiciones circulares, pero su mayor inconveniente es que el uso de memoria se vuelve muy difícil de predecir, por ejemplo las expresiones `3+2 :: Int` y `5 :: Int` denotan el mismo valor pero pueden tener diferentes tamaños en memoria.

### 4. Garbage Collection

Muchos lenguajes implementan un Garbage Collection (GC) para ocuparse de los objetos no alcanzables y liberar dicha memoria.

Los lenguajes imperativos suelen generar una cantidad de basura a lo largo de su ejecución, y si estos constan de un GC simplifican la vida del programador, al desligarlo de la necesidad de liberar el espacio no utilizado, y de no deber perder nunca la referencia. De lo contrario, si un lenguaje posee un GC, el programador debe olvidarse de lo que no necesite, simplemente haciéndolo inalcanzable, de este modo las referencias almacenadas son solo las que en ese momento resultan útiles, el GC pasará en algún momento y liberará la memoria no alcanzable.

Los lenguajes funcionales, como lo es Haskell, suelen generar muchísima más basura que los lenguajes imperativos, esto es debido a la inmutabilidad de sus variables, y para guardar un nuevo valor es necesario

crear una nueva variable. Cada iteración en un llamado recursivo genera un nuevo valor. En Haskell no es insólito producir 1Gb de datos en un segundo. Para esto el compilador de Haskell GHC tiene un potente GC que se ocupa de gestionar la basura en forma eficiente.

La inmutabilidad de las variables en Haskell no solo obliga a generar gran cantidad de basura, sino que esta característica es aprovechada para realizar la recolección. Para esto utiliza el criterio de generación: datos jóvenes y viejos. Las variables mas viejas, al ser inmutables, no apuntan nunca a un dato joven, pues los datos jóvenes no existen en el momento en que las variables viejas son creadas. Esta es la idea que utiliza el recolector de basura de Haskell para aumentar la eficiencia. El recolector de basura no mirará toda la memoria en la que estuvo trabajando nuestro programa, sino que sólo revisará entre los valores más jóvenes, y liberará los que no estén señalados, que suelen ser la gran mayoría, dado el comportamiento recursivo. Esto también es una ventaja, porque en realidad, mientras mas basura joven aparezca menor es el trabajo que realiza el recolector de basura; esto, que resulta tan poco intuitivo se explica con la forma que se utiliza para almacenar los datos jóvenes y los viejos. Los datos jóvenes son almacenados en un bloque de memoria especial, una “guardería”, cuando esta guardería se llena el GC mira sólo en esta memoria quienes son alcanzables, por lo tanto útiles y los copia en la memoria de las variables mas viejas, luego nos habilita a reutilizar la “guardería”, la cual se encuentra “vacía”, ya que cualquier dato que pisemos no será útil, o lo tendremos copiado con los valores mas viejos. Este es el motivo por el cual, con mayor cantidad de basura joven, la recolección es más rápida, lo que sucede es que hay menos datos que copiar.

## 5. Funciones de orden superior

Una función de orden superior es aquella que puede tomar funciones como parámetro, o devolver una función como resultado, o ambas cosas.

Haskell no solo soporta las funciones de orden superior, sino que hace un uso permanente de esta cualidad, y de forma muy natural.

### 5.1. Funciones como parametro

Funciones como parámetro

Una función que puede tomar como parámetro a otra función es considerada de orden superior. En Haskell esto se utiliza todo el tiempo para filtrar datos mediante algún criterio, realizar una acción sobre un conjunto de datos, etc.

Un ejemplo de una función que viene por defecto en Haskell que toma funciones como parámetro es la función `filter`, a la cual se le pasa una función y una lista, `filter` llama a la función con cada elemento de la lista, si la función devuelve `true` el elemento es añadido a la lista que da como resultado.

Si se implementa una función `fQuickSort` a la que le paso una función `f` que dado un elemento de la lista me devuelve un valor comparable, y una lista que quiero ordenar de la siguiente forma:

```

1 fQuickSort :: (Ord b) => (a -> b) -> [a] -> [a]
2 fQuickSort _ [] = []
3 fQuickSort f (x:xs) =
4   let
5     menores = fQuickSort f [a | a <- xs, (f a) <= (f x)]
6     mayores = fQuickSort f [a | a <- xs, (f a) > (f x)]
7   in menores ++ [x] ++ mayores

```

En esta implementación es realmente versátil, ya que dependiendo que función `f` se utilice para medir el “tamaño” de los elementos de la lista obtendremos un resultado distinto, y no restringimos los elementos de la lista a elementos ordenables, dado que el criterio de orden los obtenemos en función de la “medida” de los elementos.

## 5.2. Funciones como resultado

Dijimos que una función también es de orden superior si puede dar como resultado otra función. Esto es útil, pues se puede crear constructores de funciones, estos constructores recibirán un valor, y devolverán una función para cada valor que tomen. Por ejemplo:

```
1 multiplicarPor :: (Num a =>) a -> (a -> a)
2 multiplicarPor x = (*) x
```

Para cada valor de  $x$ , `multiplicarPor` da como resultado una función que recibe un número y devuelve el producto del número pasado con  $x$ .

En realidad, Haskell utiliza las funciones de orden superior todo el tiempo, ya que las funciones de Haskell solo pueden tomar una única variable, esto es lo mismo que decir que las funciones en Haskell están Currificadas. No hay ninguna contradicción, cuando tenemos una función que aparenta recibir más de una variable lo que en realidad tenemos es una función que recibe un dato y nos devuelve una función, la cual toma el siguiente dato, y nos devuelve otra función y así sucesivamente. Esta es la explicación del porque cuando anotamos en la definición de tipos de una función no diferenciamos entre los parámetros de entrada y el valor de retorno.

La Currificación hace evidente la necesidad de que el lenguaje soporte funciones de orden superior. A medida que vamos aplicando parcialmente la función vamos obteniendo nuevas funciones como resultado, y este es el concepto de que una función sea de orden superior por devolver una función. Usando el ejemplo anterior, el `fQuickSort` es un constructor de sorts, al que se le pasa una función  $f$  y nos da un sort que ordena con un determinado criterio, lo mismo pasa con `filter` y `map`.

En Haskell este tipo de funciones son tan comunes que tiene montones de aplicaciones en las librerías standar utilizando funciones de orden superior, ya mencionamos las funciones `filter` y `map`, pero se incluyen muchos más.

## 5.3. Plieges (folds)

Unas funciones particularmente útiles y cómodas son los pliegues (folds). Como en Haskell las variables son inmutables no existen los iteradores clásicos de los lenguajes imperativos, sino que se utiliza la recursividad de las funciones, esto es tan común que existen algunas funciones útiles tienen incorporado estos patrones para realizar iteraciones. Si quierámos implementar la función `elem` utilizando pliegues podríamos hacerlo como

```
1 elem' :: (Eq a) => a -> [a] -> Bool
2 elem' y ys = foldl f False ys
3 where f acc x = if x==y then True else acc
```

Lo que estamos haciendo es pasarle a nuestro pliegue una función que devuelve `True` si el elemento que recibe es el mismo que el que nosotros buscamos, y sino devuelve el `acc`, además la función recibe como “segundo” parámetro el valor inicial del acumulador que va a pasarle a la función, y una lista sobre la que deseamos que itere. Lo que hace `foldl` es agarrar la cabeza de la lista, pasárselo a la función  $f$ , y tomar el valor obtenido como nuevo acumulador, luego se llama a sí mismo con la cola de la lista, la misma función y este nuevo acumulador. El resultado que devuelve `foldl` es el último valor que devuelve el acumulador.

Si quisiéramos implementar `foldl` podríamos hacerlo así:

```
1 foldl' :: (a -> b -> a) -> a -> [b] -> a
2 foldl' _ acc [] = acc
3 foldl' f acc (x:xs) = foldl' f (f acc x) xs
```

## 5.4. Funciones Anónimas (lambdas)

Al usar este tipo de funciones solemos tener que crear funciones con el único objetivo de pasarlas a nuestras funciones de orden superior, lo cual no tiene mucho sentido, para esto aparecen las denominadas

funciones anónimas, o funciones lambdas. Las funciones lambda son expresiones (devuelven un valor), por eso podemos pasarlas como parámetros a funciones de orden superior. Su sintaxis es:

```
1 (\a b -> 2a/b)
```

donde a y b son los parámetros que recibe, lo que sucede a -> indica que el comportamiento de dicha función. Suelen estar encerradas entre paréntesis para delimitarlas, de lo contrario tomarán todo el renglón. Utilizando este tipo de funciones, podríamos haber escrito nuestro pliegue como

```
1 elem' :: (Eq a) => a -> [a] -> Bool
2 elem' y ys = foldl (\acc x-> if x==y then True else acc) False ys
```

## 6. Functores

Los funtores son una clase de tipos que tiene un método llamado fmap, donde

```
1 fmap :: (a -> b) -> f a -> f b
```

La idea de fmap es que toma una función de a en b, un functor que contiene un a, y devuelve un functor que tiene un b. Una idea intuitiva de un functor podría ser como una caja, fmap abre esta caja, aplica f al contenido y devuelve una caja con el resultado de aplicar la función.

Un término más correcto para definir lo que es un functor sería contexto computacional. El contexto sería que la computación podría tener un valor, o podría fallar. Si queremos un constructor de tipos que sea una instancia de functor este debe pertenecer a la familia  $* \rightarrow *$ .

Algunos funtores básicos son las listas [], Maybe, Either a (en este último caso, como necesitamos que nuestro tipo tome un único tipo concreto y Either toma dos debe estar parcialmente evaluado), si se definiese una clase Tree a también podría ser un functor. Otros no tan claros son IO y (->) r.

Las acciones IO son como cajas que encierran datos que provienen del mundo real, o que saldrán al mismo, podemos ligar el contenido de una acción IO a una variable usando <-, trabajar con estos datos, realizar algunos cálculos, pero cuando queramos sacarlos al mundo exterior es necesario transformarlos en otra acción IO usando return. Si queremos mapear un valor IO lo que en realidad buscamos es obtener una nueva acción IO que contenga el resultado de aplicar una determinada función, esto será:

```
1 instance Functor IO where
2   fmap f action = do
3     result <- action
4     return (f result)
```

El tipo de una función  $r \rightarrow a$  se puede reescribir como  $(\rightarrow) r$  a esto indica que  $(\rightarrow) r$  es una caja de a. Como  $\rightarrow$  es de la familia  $* \rightarrow * \rightarrow *$  debe estar parcialmente aplicado, igual que Either.

```
1 instance Functor ((->) r) where
2   fmap f g = (\x -> f (g x))
```

Aplicar fmap a una función nos da una nueva función, si pensamos que aplicamos esta nueva función a un valor esto daría por resultado la función original aplicada a nuestro valor, y a este resultado le aplicaría la función que le pasamos a fmap, esto no es otra cosa que realizar la composición de funciones.

```
1 instance Functor ((->) r) where
2   fmap = (.)
```

### 6.1. Leyes de los Funtores

Ahora vamos a ver las leyes de los funtores. Para que algo sea un functor, debe satisfacer una serie de leyes. Se espera que todos los funtores exhiban una serie de propiedades y comportamientos. Deben comportarse fielmente como cosas que se puedan mapear. Al llamar fmap sobre un functor solo debe mapear una función sobre ese functor, nada más. Este comportamiento se describe en las leyes de los



funtores. Hay dos de ellas que todas las instancias de `Functor` deben cumplir, pero Haskell no hace esta comprobación, debe ser verificada por el usuario que la implementa.

La primera ley de funtores establece que si mapeamos la función identidad sobre un functor, el functor que obtenemos debe ser igual que el original (`fmap id = id`).

La segunda ley dice que si mapeamos el resultado de una composición de dos funciones sobre un functor debe devolver lo mismo que si mapeamos una de estas funciones sobre el functor inicial y luego mapeamos la otra función (`(f . g) = fmap f`).

## 6.2. Funtores Aplicativos

Dentro de un functor podemos tener almacenado cualquier tipo de valor, y como en Haskell las funciones son ciudadanos de primer orden estas también son un valor, por ende podemos tener funciones encerradas dentro de un functor, por ejemplo, si hacemos

```
1 fmap (*) Just 7
```

obtenemos

```
1 Just (* 7)
```

Si queremos aplicar la función almacenada a el valor almacenado por `Just 11` podríamos tratar de hacerlo, pero si queremos un comportamiento que funcione en todos los funtores debemos crear algo más general, este es el motivo por el cual aparecen los funtores aplicativos.

```
1 class (Functor f) => Applicative f where
2   pure :: a -> f a
3   (<*>) :: f (a -> b) -> f a -> f b
```

La definición de esta clase de tipos nos dice que si alguien es un `Applicative` también es un `Functor`. Las funciones no están implementadas por defecto, pero `pure` toma un valor y lo encapsula dentro de un functor aplicativo. La segunda función es una ampliación de `fmap`, toma un functor aplicativo, “extrae” su función, y mapea dicha función sobre otro functor. La clase de tipos `Maybe` los implementa de la siguiente forma:

```
1 instance Applicative Maybe where
2   pure = Just
3   Nothing <*> _ = Nothing
4   (Just f) <*> something = fmap f something
```

## 7. Mónadas

Las **mónadas** en Haskell se pueden pensar como descripciones *componibles* de computaciones. Presentan la posibilidad de separar la combinación de computaciones de su ejecución y permiten acarrear datos extra implícitamente en adición al resultado de la computación, que *se producirá* cuando la mónada sea corrida. De esta manera permiten suplementar las funcionalidades *puras* con I/O, estado, indeterminismo, etc.

En términos del lenguaje una mónada es un tipo parametrizado que es instancia de la clase *Monad*. Su definición es la siguiente:

```
1 class Monad m where
2   return :: a -> m a
3   (>>=) :: m a -> (a -> m b) -> m b
4   (>>) :: m a -> m b -> m b
```

Podemos ver la mónada como un contenedor para un valor **a**. La función `return` se ocupa de poner ese valor adentro de la mónada. Entonces la función `(>>=)`, también conocida como *bind*, aplica la

función que se le pasa por parámetro al contenido de la mónada obteniendo como resultado otra mónada (obviamente la función pasada tiene que tener el tipo adecuado). Se puede ver como funciona en el siguiente ejemplo:

```

1
2 putStrLn "Como te llamas?"
3 >>= (\_ -> getLine)
4 >>= (\name -> putStrLn ("Hola, " ++ name ++ "!!"))

```

El operador ( $>>=$ ) se ocupa de tomar el valor del lado izquierdo y combinarlo con la función del lado derecho para producir un valor nuevo. El ejemplo de arriba se puede reescribir con la notación **do**, que es un azúcar sintáctico alrededor del operador *bind*:

```

1
2 do
3   putStrLn "Como te llamas?"
4   name <- getLine
5   putStrLn ("Hola, " ++ name ++ "!!")

```

Ese código puede parecer de un lenguaje imperativo, y de hecho lo es: otra forma de ver las mónadas es pensar que son la abstracción necesaria para suplementar las funcionalidades que no cuadran adentro del paradigma funcional.

La implementación mas sencilla del ( $>>=$ ), toma el valor del lado izquierdo, le aplica la función y devuelve el resultado, sin embargo se vuelve realmente útil cuando esa implementación hace algo extra.

## 7.1. Las Leyes de las Mónadas

Las mónadas por convención deben cumplir las siguientes leyes:

```

1 -- Identidad por la izquierda
2 return x >>= f = f x
3
4 -- Identidad por la derecha
5 m >>= return = m
6
7 -- Asociatividad
8 (m >>= f) >>= g = m >>= (x -> f x >>= g)

```

Para poder entenderlo mejor y usarlo de una forma más sencilla, el módulo Control.Monad define el operador de composición de mónadas:

```

1 (>=>) :: Monad m => (a -> m b) -> (b -> m c) -> a -> m c
2 (m >=> n) x = do
3   y <- m x
4   n y

```

Con el uso de ese operador las tres reglas pueden escribirse de la siguiente forma:

```

1 -- Identidad por la izquierda
2 return >=> f = f
3
4 -- Identidad por la derecha
5 f >=> return = f
6
7 -- Asociatividad
8 (f >=> g) >=> h = f >=> (g >=> h)

```

## 7.2. Mónadas comunes

La siguiente tabla lista las mónadas más comunes usadas en Haskell, denotando el problema que tratan de solucionar en términos imperativos.

Mónada	Semántica imperativa
Maybe	Excepción anónima
Error	Excepción con descripción
State	Estado global
IO	Entrada y Salida
[] (list)	Indeterminismo
Reader	Entorno
Writer	Logger

En las siguientes secciones se explicarán algunas de las mónadas que aparecen en esta tabla.

## 7.3. Mónada Maybe

Es una de las mónadas más utilizadas y es muy sencilla a la vez. La definición de esta Mónada es la siguiente:

```

1 data Maybe a = Nothing | Just a
2
3 instance Monad Maybe where
4     return      = Just
5     fail        = Nothing
6     Nothing >>= f = Nothing
7     (Just x) >>= f = f x

```

La mónada Maybe incorpora la posibilidad de encadenar computaciones que pueden devolver Nothing como resultado, en cuyo caso la cadena terminaría antes.

### 7.3.1. Either

Otra mónada muy usada es *Either*. Su tipo está definido de la siguiente manera:

```

1 data Either a b = Left a | Right b

```

Su propósito es muy parecido al de Maybe, Left es considerado un error mientras que Right - un valor normal. La diferencia es que Left permite guardar un valor (a diferencia del Nothing en el caso de Maybe).

## 7.4. Mónada Error

La mónada Error es la forma que tiene Haskell de simular las *excepciones*. Se logra de la siguiente forma: se encadenan computaciones que pueden lanzar excepción derivando la ejecución a la instancia que puede manejarla.

## 7.5. Mónada List

Resulta que la lista es una mónada también! Las listas se utilizan para modelar las computaciones no determinísticas que pueden devolver un número de resultados arbitrario. Se define de la siguiente manera:

```

1 instance Monad [] where
2     m >>= f = concat (map f m)
3     return x = [x]

```

El `return` es fácil de entender: simplemente devuelve una lista de un elemento. El operador `bind` también se entiende si consideramos que tiene que cumplir con el tipo  $[a] \rightarrow (a \rightarrow [b]) \rightarrow [b]$  (por su definición más general). La función `f` aplica a cada elemento `y`, dado su tipo, retorna una lista - por eso es necesario el `concat` al final, para a la salida obtener una lista.

¿Por qué las listas son mónadas? Se explica con el siguiente ejemplo de notación monádica de una lista:

```
1 foo = do
2   x <- [1 .. 10]
3   y <- [2, 3, 5, 7]
4   return (x * y)
```

`foo` es un múltiplo de `x` e `y`, con `x` siendo un número no determinístico entre 1 y 10, e `y` siendo 2, 3, 5 o 7. Este ejemplo es la notación larga y monádica de una comprensión de lista, podría ser escrita así:

```
1 foo = [x * y | x <- [1 .. 10], y <- [2, 3, 5, 7]]
```

## 7.6. Mónada IO

Dado que Haskell es un lenguaje funcional y perezoso, no podemos expresar los efectos reales de las operaciones de entrada y salida con funciones puras. De hecho, estas operaciones no se pueden ejecutar perezosamente, ya que esto haría que los efectos reales sean impredecibles. La mónada **IO** es el medio para representar las acciones de entrada/salida como valores de Haskell, para que el programador pueda manipularlos con funciones puras.

## 7.7. Mónada State

La mónada *State* permite acarrear estado a lo largo de una ejecución. Con ella se puede hacer lo siguiente: dado un valor de estado, se produce un resultado y un nuevo valor de estado. Esta es su definición:

```
1 newtype State s a = State { runState :: (s -> (a,s)) }
2
3 instance Monad (State s) where
4   return x = State $ \s -> (x,s)
5   (State h) >>= f = State $ \s -> let (a, newState) = h s
6                                   in  (State g) = f a
7                                   in  g newState
```

Como se ve en la declaración de tipo, `State` es solo una abstracción de una función que toma un estado, devuelve un valor intermedio y un nuevo estado.

Un ejemplo de uso de `State` es una pila. La función **push** agrega un elemento al tope de la pila y **pop** saca uno. Sin `State` tendríamos que arrastrar la pila como argumento a estas funciones, lo cual no es lo más cómodo de usar. Con `State` la podemos definir de la siguiente forma:

```
1 type Stack = [Int]
2
3 pop :: State Stack Int
4 pop = State $ \(x:xs) -> (x,xs)
5
6 push :: Int -> State Stack ()
7 push a = State $ \xs -> ((),a:xs)
```

Un ejemplo de uso de la pila sería:

```
1 stackManip = do
2   push 3
3   a <- pop
4   pop
```

La implementación de State en Control.Monad también es instancia de la clase MonadState que define 2 funciones muy útiles:

```
1 put newState = State $ \_ -> ((), newState)
2 get = State $ \st -> (st, st)
```

Las cuales nos permiten manejar el estado de una manera más sencilla.

## 8. Zippers

Los zippers tienen como objetivo lograr un movimiento de forma eficiente en una estructura, conocida, que almacena datos. En un zipper se almacena el dato al que se ha llegado por un movimiento y, además, la información necesaria para reconstruir toda la estructura original.

Un problema muy sencillo para ver la utilidad de los zippers es el de querer aplicar una función en una hoja de un árbol, y luego aplicar la misma función en el hermano de dicha hoja. Si no se contase con una estructura del tipo zipper, habría que realizar todo el recorrido hasta la hoja, luego aplicar la función, y luego volver, desde la raíz a realizar el mismo recorrido, salvo en el último movimiento, y nuevamente aplicar la función. Usando los zippers podemos movernos hasta la hoja deseada, y como podemos reconstruir la estructura, podemos pedir el padre, y además saber si el nodo actual era el hijo izquierdo o el derecho, para así saltar al nodo hermano y aplicar nuevamente la función. Con este ejemplo se observa que al querer realizar determinadas acciones con bloques de información de una estructura, y estos datos se encuentran “próximos” en dicha estructura, al construir un zipper para esta estructura nos ahorramos recorrer reiteradas veces la estructura original, mejorando la velocidad de nuestro programa.

Volviendo al ejemplo del árbol binario podemos definir las estructuras de la siguiente forma:

```
1 data Tree a = Empty | Node a (Tree a) (Tree a) deriving (Show)
2 data Crumb a = LeftCrumb a (Tree a) | RightCrumb a (Tree a) deriving (Show)
3 type Breadcrumbs a = [Crumb a]
4
5 type Zipper a = (Tree a, Breadcrumbs a)
```

Aquí se crea el tipo árbol, el cual puede estar vacío o tener un dato y otros dos árboles. El tipo miga, el cual dice si nos movimos a la derecha o a la izquierda, contiene la información que contenía el nodo padre, y además tiene el subárbol no visitado. Además se crea el sinónimo de tipo migas de pan, que simplemente es una lista de migas. Por ultimo se crea el tipo zipper, el cual tiene el árbol en el cual estaremos parados, y además las migas de pan del camino recorrido. Con todos estos tipos de datos definidos podemos crear funciones que nos permitan movernos por de forma adecuada por los árboles.

```
1 goUp :: (Tree a, Breadcrumbs a) -> (Tree a, Breadcrumbs a)
2 goUp (t, LeftCrumb x r:bs) = (Node x t r, bs)
3 goUp (t, RightCrumb x l:bs) = (Node x l t, bs)
4
5 goRight :: (Tree a, Breadcrumbs a) -> (Tree a, Breadcrumbs a)
6 goRight (Node x l r, bs) = (r, RightCrumb x l:bs)
7
8 goLeft :: (Tree a, Breadcrumbs a) -> (Tree a, Breadcrumbs a)
9 goLeft (Node x l r, bs) = (l, LeftCrumb x r:bs)
```

Con estas tres funciones podemos movernos libremente, y eficientemente, sobre el árbol. Si una vez alcanzada una determinada posición queremos realizar un cambio sobre ese árbol podemos usar la siguiente función

```
1 modify :: (a -> a) -> Zipper a -> Zipper a
2 modify f (Node x l r, bs) = (Node (f x) l r, bs)
3 modify f (Empty, bs) = (Empty, bs)
```

Esta función recibirá una función y un zipper, y nos devolverá un nuevo zipper, el cual es el resultado de aplicarle la función al nodo actual. Además podrían crearse otras funciones útiles, por ejemplo, una

función que reciba un árbol y devuelva un zipper de dicho árbol, utilizando el árbol pasado como la raíz, una función que dado un zipper indique si se puede subir, o estamos en la raíz del árbol, otra que agregue/reemplace un subárbol por otro que le pasemos, otra que se ocupe de llegar a la raíz, etc.

Otra estructura, más sencilla, en la que son útiles los zippers son las listas, donde los movimientos posibles serían avanzar o retroceder:

```

1 type ListZipper a = ([a],[a])
2
3 goForward :: ListZipper a -> ListZipper a
4 goForward (x:xs, bs) = (xs, x:bs)
5
6 goBack :: ListZipper a -> ListZipper a
7 goBack (xs, b:bs) = (b:xs, bs)

```

La primera lista del zipper es la cola de la lista original que resulta del movimiento, la segunda es una lista invertida de los elementos que están por delante del elemento actual. En este caso los zippers son mas simples que en los árboles, pero no por esto dejan de ser útiles. En Haskell los Strings son un sinónimo de tipo de una lista de caracteres, o sea que podemos usar un zipper para movernos dentro de un texto, y modificar y o agregar caracteres en el mismo, esto es realmente útil si estamos creando un editor de texto.

## 9. Concurrency

Un programa concurrente necesita realizar varias tareas al mismo tiempo. Estas tareas no necesariamente tienen que estar relacionadas entre si. El correcto funcionamiento de un programa concurrente no necesita varios núcleos.

En contraste, un programa paralelo soluciona un solo problema con el mejor rendimiento posible, empleando para eso más de un núcleo.

### 9.1. Threads

Un hilo es una acción *IO* que se ejecuta independientemente de los otros hilos. Los hilos en Haskell no son determinísticos. Para crear un thread, usamos la función *forkIO* del módulo *Control.Concurrent*.

Un ejemplo de uso podría ser la compresión de un archivo:

```

1 import Control.Concurrent (forkIO)
2 import qualified Data.ByteString.Lazy as L
3 import Codec.Compression.GZip (compress)
4
5 compressFile = L.writeFile "files/foo.gz" . compress
6
7 do
8   content <- L.readFile "files/foo.txt"
9   forkIO (compressFile content)
10  putStrLn "Gracias por comprimir!"
11  return ()

```

### 9.2. Comunicación entre hilos

La forma más simple que tiene un hilo para comunicarse con otro es compartiendo una variable. En Haskell podemos lograrlo utilizando una variable sincronizable **MVar**. Una **MVar** actúa como una caja con un solo elemento, que puede estar vacía o llena. Si tratamos de poner un valor adentro de una **MVar** que ya está llena, la ejecución del hilo en cuestión se suspenderá hasta que otro hilo la vacíe. De la misma manera, si intentamos sacar el valor de una **MVar** vacía, el hilo se suspenderá hasta que algún otro thread le agregue algo. En el siguiente ejemplo se puede ver el uso de una **MVar**:

```

1 import Control.Concurrent
2
3 do
4   m <- newEmptyMVar
5   forkIO $ do
6     v <- takeMVar m
7     putStrLn ("Recibido " ++ show v)
8   putStrLn "Enviando"
9   putMVar m "Hola!"

```

Utilizamos la función *newEmptyMVar* para crear una MVar, *putMVar* para ponerle un valor adentro y *takeMVar* para sacarlo. Notese que solo podemos utilizar una MVar adentro de un bloque *do*.

### 9.3. Comunicación por canales

Utilizando el tipo *Chan*, podemos crear una comunicación mediante un canal. El siguiente código ejemplifica su uso:

```

1 import Control.Concurrent
2 import Control.Concurrent.Chan
3
4 do
5   ch <- newChan
6   forkIO $ writeChan ch "Hola!"
7   forkIO $ writeChan ch "Hola desde otro hilo!"
8   readChan ch >>= print
9   readChan ch >>= print

```

Si el *Chan* está vacío, el bloque *readChan* espera hasta que haya un valor para leer. La función *writeChan* nunca se bloquea, escribe al canal inmediatamente.

### 9.4. Aclaraciones y problemas

Para tener en cuenta: **MVar** y **Chan** no son *estrictos*, su contenido no es evaluado sin explicitarlo. Además, el **writeChan** siempre acepta lo que se le escribe. Si un hilo escribe más frecuentemente que otro lee los mensajes, el canal crecerá desatendidamente.

Ambos métodos de comunicación entre hilos sufren de una serie de problemas típicos de concurrencia de estado compartido (en el caso de *Chan* se debe a que internamente están implementados con Mvars). Los problemas más conocidos son *deadlock* y *starvation*. En una situación de *Deadlock* dos o más hilos quedan suspendidos para siempre esperando el acceso a un recurso compartido. *Starvation* ocurre cuando un recurso es tomado por mucho tiempo por un hilo, mientras otro hilo no puede progresar, esperando el acceso (y probablemente la ejecución de ese último hilo es mucho más rápida).

Esos problemas típicos se tratan de solucionar con formas más recientes de hacer concurrencia, como por ejemplo *Software Transactional Memory*, que analizaremos más adelante.

## 10. Paralelismo

Para muchos problemas costosos de resolver, podríamos obtener el resultado más rápido separando la solución y evaluandola en varios núcleos. Por defecto ghc utiliza un solo núcleo. Para poder tomar provecho de múltiples núcleos necesitamos pasarle la opción **-threaded** al compilador a la hora de linkear.

Un ejemplo de uso de evaluación paralela sería la siguiente función, inspirado en el famoso algoritmo Quicksort. El siguiente código es el sort secuencial:

```

1 sort :: (Ord a) => [a] -> [a]
2 sort (x:xs) = lesser ++ x:greater
3   where lesser = sort [y | y <- xs, y < x]

```

```

4     greater = sort [y | y <- xs, y >= x]
5 sort _ = []

```

De la siguiente manera lo convertimos en un código que se ejecuta en paralelo.

```

1 import Control.Parallel (par, pseq)
2
3 parSort :: (Ord a) => [a] -> [a]
4 parSort (x:xs) = force greater 'par' (force lesser 'pseq'
5                                     (lesser ++ x:greater))
6     where lesser = parSort [y | y <- xs, y < x]
7           greater = parSort [y | y <- xs, y >= x]
8 parSort _ = []

```

Es solo un poco más complicado que la versión inicial: agregamos las funciones **par**, **pseq** y **force**. La función **par** evalúa el argumento de la derecha a *Weak Head Normal Form* y retorna el de la izquierda. Para programas paralelos *Weak Head Normal Form* significa que la expresión está evaluada hasta el constructor de más afuera. Por ejemplo, para la expresión **[1, 2, 3]**, *WHNF* es evaluar el constructor de lista y no el contenido de la misma. Lo interesante del **par**, es que *puede* evaluar el argumento a la izquierda en paralelo.

La función **pseq** es similar, solo que garantiza que el elemento a la izquierda será evaluado a *WHNF* antes del argumento de la derecha. **force** es necesario para forzar la evaluación de la lista. Podemos definir **force** de la siguiente manera:

```

1 force :: [a] -> ()
2 force xs = go xs 'pseq' ()
3     where go (_:xs) = go xs
4           go [] = 1

```

La función **par** no promete que una expresión será evaluada en paralelo con otra, pero si promete hacerlo si tiene sentido. Esta promesa le da la posibilidad de actuar inteligentemente a la hora de ejecutar un **par**. En tiempo de ejecución, el programa puede decidir que la expresión es muy liviana, como para que la ejecución paralela tenga sentido. O puede notar que todos los núcleos están ocupados en el momento.

## 11. Software Transactional Memory

*Software Transactional Memory*, **STM**, es un módulo que nos da herramientas simples, pero poderosas, que buscan solucionar los problemas de concurrencia comunes (los mencionamos en el capítulo 9).

La idea es muy sencilla: ejecutamos un bloque de acciones como una transacción, utilizando la función **atomically**. Una vez que entramos a ese bloque, otros hilos no pueden ver ninguna modificación hasta que terminemos la ejecución, y tampoco podemos ver modificaciones realizadas por otros hilos desde ese bloque. Por esas propiedades decimos que nuestra ejecución está *isolada*.

Cuando salimos de la transacción, puede ocurrir una de las dos posibilidades:

1. Si ningún otro hilo modifica los mismos datos, todas las modificaciones ocurridas durante la transacción serán visibles para los otros threads.
2. En el otro caso, todas las modificaciones hechas se descartan y nuestro bloque de acciones se vuelve a ejecutar automáticamente.

En el siguiente ejemplo sencillo de una cuenta bancaria podemos ver su funcionamiento:

```

1 import Control.Concurrent.STM
2
3 type Account = TVar Int
4
5 withdraw :: Account -> Int -> STM ()
6 withdraw acc amount = do
7     bal <- readTVar acc

```



```
8   writeTVar acc (bal - amount)
9
10 deposit :: Account -> Int -> STM ()
11 deposit acc amount = withdraw acc (- amount)
12
13 atomically $ do
14   acc <- newTVar (20 :: Int)
15   deposit acc 20
16   withdraw acc 10
17   readTVar acc
```

## Referencias

- [1] ¡Aprende Haskell por el bien de todos!, <http://aprendehaskell.es>
- [2] Real World Haskell, <http://book.realworldhaskell.org>
- [3] “What part of Milner-Hindley do you not understand?”, <http://stackoverflow.com/questions/12532552/what-part-of-milner-hindley-do-you-not-understand>
- [4] Recursion, <http://en.wikibooks.org/wiki/Haskell/Recursion>
- [5] Recursion Patterns, <https://www.fpcomplete.com/school/starting-with-haskell/introduction-to-haskell/3-recursion-patterns-polymorphism-and-the-prelude>
- [6] GHC Memory Management, [https://wiki.haskell.org/GHC/Memory\\_Management](https://wiki.haskell.org/GHC/Memory_Management)
- [7] Laziness, <http://en.wikibooks.org/wiki/Haskell/Laziness>
- [8] Understanding Monads, [http://en.wikibooks.org/wiki/Haskell/Understanding\\_monads](http://en.wikibooks.org/wiki/Haskell/Understanding_monads)
- [9] Monad, <https://wiki.haskell.org/Monad>
- [10] What is a monad?, <http://stackoverflow.com/questions/44965/what-is-a-monad>
- [11] Monad laws, [https://wiki.haskell.org/Monad\\_laws](https://wiki.haskell.org/Monad_laws)
- [12] All about monads, [https://wiki.haskell.org/All\\_About\\_Monads](https://wiki.haskell.org/All_About_Monads)
- [13] The State Monad: A Tutorial for the Confused?, <http://brandon.si/code/the-state-monad-a-tutorial-for-the-confused/>